



Maxima 5.27.0 <http://maxima.sourceforge.net>
 using Lisp GNU Common Lisp (GCL) GCL 2.6.7 (a.k.a. GCL)
 Distributed under the GNU Public License. See the file COPYING.
 Dedicated to the memory of William Schelter.

Distance et perpendiculaire commune de deux droites

On commence par charger le fichier contenant la définition de deux macros :

PointDeLaDroite – détermination d'un point d'une droite définie comme intersection de deux plans,

VecteurDeLaDroite – détermination d'un vecteur directeur d'une droite définie comme intersection de de deux plans.

Ce même fichier charge `vect.mac` qui contient la définition de `~` servant à calculer un produit vectoriel.

```
▷ load("geo3d.mac")$
```

On introduit la première droite \mathcal{D}_1 .

```
▷ D1:[x=3*z+1,y=2*z-1];
2:      [x = 3z + 1, y = 2z - 1]
```

Puis \mathcal{D}_2 .

```
▷ D2:[y=x-2,z=1];
3:      [y = x - 2, z = 1]
```

On détermine un point et un vecteur directeur de \mathcal{D}_1 .

```
▷ block(A1:PointDeLaDroite(D1,z=0),V1:VecteurDeLaDroite(D1,z=0),[A1,V1]);
4:      [[1, -1, 0], [3, 2, 1]]
```

On en fait autant pour \mathcal{D}_2 .

```
▷ block(A2:PointDeLaDroite(D2,x=0),V2:VecteurDeLaDroite(D2,x=0),[A2,V2]);
5:      [[0, -2, 1], [1, 1, 0]]
```

On détermine, à l'aide du produit vectoriel, un vecteur directeur de la perpendiculaire commune Δ .

```
▷ V:express(V1~V2);
6:      [-1, 1, 1]
```

On introduit un point générique.

```
▷ M:[x,y,z];
7:      [x, y, z]
```

On détermine une équation du plan contenant \mathcal{D}_1 et Δ .

```
▷ P1:expand((M-A1).express(V~V1))=0;
8:      -5z + 4y - x + 5 = 0
```

Maintenant le plan contenant \mathcal{D}_2 et Δ .

▷ `P2:expand((M-A2).express(V~V2))=0;`

9: $-2z + y - x + 4 = 0$

Nous pouvons définir Δ comme intersection des deux plans précédents.

▷ `Delta:[P1,P2];`

10: $[-5z + 4y - x + 5 = 0, -2z + y - x + 4 = 0]$

À titre d'exemple voici un point de Δ .

▷ `A:PointDeLaDroite(Delta,y=x-2);`

11: $\left[\frac{8}{3}, \frac{2}{3}, 1 \right]$

Et la distance des deux droites \mathcal{D}_1 et \mathcal{D}_2 .

▷ `d:abs((A2-A1).V)/sqrt(V.V);`

12: $\frac{1}{\sqrt{3}}$

geo3d.mac

```
load("vect.mac");

PointDeLaDroite(d,e) := block([s],
  s:part(solve(append(d,[e]),[x,y,z]),1),
  [rhs(part(s,1)),rhs(part(s,2)),rhs(part(s,3))]
);

VecteurDeLaDroite(d,e) := block([p1,p2],
  p1:PointDeLaDroite(d,e),
  p2:PointDeLaDroite(d,e+(0=1)),
  p2-p1
);
```